
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of NIMTC testcase 6.1.1.7a which is part of the LTE_A test suite in the ‘iwd-TTCN3-B2013-03_D14wk37’ delivery.

The test case can be demonstrated to run with one LTE UE supporting NIMTC (see section 5). Execution logs are provided as evidence.

3. Verification Test Summary

Test Case:
TC_6_1_1_7a
Test Group:
LTE_A_IWD_14wk37
ATS Version:
iwd-TTCN3-B2013-03_D14wk37
System Simulator used:
Anritsu Protocol Conformance Test System ME7832L
UE used:
Intel XMM7260
Verification Status:
PASS


4. Corrections required for NIMTC test case 6.1.1.7a
This section describes the TTCN changes required to ‘iwd-TTCN3-B2013-03_D14wk37’ release
Change 1
	Function name
	TC_6_1_1_7a()

	Reason for change
	This Testcase contains several timers that extend the execution time for this TC. Guardtimer duration used in similar TC 6.1.1.7 in duration 1200 seconds should be sufficient also for TC 6.1.1.7a.

	Summary of change
	Extended guard timer to 1200 seconds.

	TTCN module
	\LTE_A_Testsuite.ttcn

	MCC160 Comment
	Accepted


Before change

  testcase TC_6_1_1_7a() runs on MTC_LTE_IRAT system SYSTEM_LTE_IRAT {

    // @purpose

    //   PLMN selection / Periodic reselection / MinimumPeriodicSearchTimer / Single Frequency operation

    var EUTRA_PTC    v_EUTRA      := null;

    var UTRAN_PTC    v_UTRAN      := null;

    var GERAN_PTC    v_GERAN      := null;

    var CDMA2000_PTC v_CDMA2000   := null;

    var IMS_PTC      v_IMS1       := null;

    var IMS_PTC      v_IMS2       := null;

    timer t_GuardTimer := int2float(360);

    v_EUTRA := EUTRA_PTC.create alive;

    f_MTC_ConnectPTCs_LTE_IRAT(v_EUTRA, v_UTRAN, v_GERAN, v_CDMA2000, v_IMS1, v_IMS2);

    v_EUTRA.start(f_TC_6_1_1_7a_EUTRA());

    t_GuardTimer.start;

    f_MTC_MainLoop(t_GuardTimer);

  }
After change

testcase TC_6_1_1_7a() runs on MTC_LTE_IRAT system SYSTEM_LTE_IRAT {

    // @purpose

    //   PLMN selection / Periodic reselection / MinimumPeriodicSearchTimer / Single Frequency operation

    var EUTRA_PTC    v_EUTRA      := null;

    var UTRAN_PTC    v_UTRAN      := null;

    var GERAN_PTC    v_GERAN      := null;

    var CDMA2000_PTC v_CDMA2000   := null;

    var IMS_PTC      v_IMS1       := null;

    var IMS_PTC      v_IMS2       := null;

    timer t_GuardTimer := int2float(1200); 

    v_EUTRA := EUTRA_PTC.create alive;

    f_MTC_ConnectPTCs_LTE_IRAT(v_EUTRA, v_UTRAN, v_GERAN, v_CDMA2000, v_IMS1, v_IMS2);

    v_EUTRA.start(f_TC_6_1_1_7a_EUTRA());

    t_GuardTimer.start;

    f_MTC_MainLoop(t_GuardTimer);

  }
Change 2
	Function name
	f_TC_6_1_1_7a_EUTRA()

	Reason for change
	As per TS 36.304 clause 5.1.2.2 (PLMN selection – E-UTRA case):

The UE shall scan all RF channels in the E-UTRA bands according to its capabilities to find available PLMNs. On each carrier, the UE shall search for the strongest cell and read its system information, in order to find out which PLMN(s) the cell belongs to. If the UE can read one or several PLMN identities in the strongest cell, each found PLMN (see the PLMN reading in [3]) shall be reported to the NAS as a high quality PLMN (but without the RSRP value) …
TC 6.1.1.7a is run in single frequency operation mode and in this case, we need to modify the cell power of the Cell 2 to be reported as the strongest cell so UE can select this cell according to TC 36.523-1 table Table 6.1.1.7a.3.2-2 Step 3-18. In current implementation the UE can only see the Cell 4 as it does not scan the weaker cell on same frequency.

Similar cell power modification needs to be done also after Step 19 as in this case the UE will also not search for weaker cell and will cause the UE not select higher priority PLMN.

Prose CR R5-145415 will be raised for these changes.

	Summary of change
	1) Modified cell powers after in step 1 (at T1)
2) Modified cell powers after in step 19A (at T2)

3) Modified cell powers after in step 27 (at T3)

	TTCN module
	\LTE_A\6_1\Idle_PLMNSelection_A.ttcn

	MCC160 Comment
	Accepted


Before change

  function f_TC_6_1_1_7a_EUTRA() runs on EUTRA_PTC

  { /* PLMN selection / Periodic reselection / MinimumPeriodicSearchTimer / Single Frequency operation */

    var template (value) CellPowerList_Type v_CellPowerList_AtT1;

    var template (value) CellPowerList_Type v_CellPowerList_AtT2;

    var EUTRA_CellId_Type v_CellId := eutra_Cell4;

    var EUTRA_ATTACH_Type v_ForcedAttach := NORMAL;

    var NAS_MSG_Indication_Type v_NasInd;

    var NAS_KsiValue v_KsiValue;

    var GutiParameters_Type v_GutiParams;

    var B3_Type v_EpsUpdate_TypeValue := f_GetEPSTAUType(v_ForcedAttach);

    var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus;

    var template AdditionalUpdateType v_AdditionalUpdateType_Expected := f_GetAdditionalUpdateType(v_ForcedAttach);

    var NAS_PlmnId v_PLMN;

    var NAS_Lac v_LAC;

    var TrackingAreaCode v_Tac;

    var template (omit)  LocAreaId v_LAI;

    var template (value) TrackingAreaIdList v_TaiList;

    var template (omit)  MobileIdentity v_MSId := f_GetMSId(v_ForcedAttach);

    timer t_HpPlmn_lower;

    timer t_HpPlmn_upper;

    timer t_IdleMode_GenericTimer := tsc_IdleMode_GenericTimer;

    v_CellPowerList_AtT1 := {

      cs_CellPower(eutra_Cell4, -79)

    };

    v_CellPowerList_AtT2 := {

      cs_CellPower(eutra_Cell1, -79),

      cs_CellPower(eutra_Cell2, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower(eutra_Cell4, tsc_ServingCellRS_EPRE)

    };
    //Initialise all cell, security and mobile parameters

    f_EUTRA_Init(c1);

    //Set maximum cell powel level for Cell 1 & 4 to be used in creation

    f_EUTRA_CellInfo_InitMaxReferencePower ( eutra_Cell1, -79 );

    f_EUTRA_CellInfo_InitMaxReferencePower ( eutra_Cell4, -79 );

    //Set Sys info and cell info for cell 2, &4 to be used @ creation

    f_EUTRA_CellInfo_SetPLMN_1Entry(eutra_Cell2, cs_HPLMN_001_11);  //PLMN1

    f_EUTRA_CellInfo_SetPLMN_1Entry(eutra_Cell4, cs_HPLMN_001_21);  //PLMN2

    //Set TAC for Cell2, &Cell4

    //TAC for Cell1 is set to 1 in f_EUTRA_Init

    f_EUTRA_CellInfo_SetTAC(eutra_Cell2, tsc_IdleMode_TAC_Cell2);

    f_EUTRA_CellInfo_SetTAC(eutra_Cell4, tsc_IdleMode_TAC_Cell4);

    //Configure and activate all cells

    //Cell power level will be -85 dBm/15kHz & attenuation set so that cell is seen as OFF

    f_EUTRA_CellConfig_Def(eutra_Cell2);

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_CellConfig_Def(eutra_Cell4);

    // initialization of variables

    v_GutiParams:= f_EUTRA_CellInfo_GetGuti(v_CellId);

    v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode(v_CellId);

    v_Tac := f_EUTRA_CellInfo_GetTAC(v_CellId);

    v_PLMN := f_Asn2Nas_PlmnId(v_GutiParams.PLMN_Identity);

    v_LAI := f_GetLAI(v_PLMN, v_LAC, v_ForcedAttach);

    v_TaiList := cds_TAIListNonConsecutive_tlv(v_PLMN, { bit2oct(v_Tac) });

    //Equip UE with a USIM containing default values except for those shown in Table 6.1.1.7a.3.1-2

    f_UT_USIM_Insert(UT, "36.523-1 Table 6.1.1.7a.3.1-2");

    //Preamble to ensure UE is in known state

    f_EUTRA_Preamble(eutra_Cell2, STATE2_IDLEUPDATE);

    f_EUTRA_UE_Detach_SwitchOffUe(eutra_Cell2, RRC_IDLE);

    f_EUTRA_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    //Adjust cell configuration according to T1

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 2" siclog@

    //Power on UE

    t_HpPlmn_lower.start(f_EUTRA_SetTimerToleranceMin(eutra_Cell2, nasTimer, 120.0));

    f_EUTRA_SwitchOnUEandStartIP(eutra_Cell2, CNF_REQUIRED);

    t_HpPlmn_upper.start(f_EUTRA_SetTimerToleranceMax(eutra_Cell2, nasTimer, 420.0));

    //@siclog "Step 3-18" siclog@

    //Steps 2 to 17 of the registration procedure described in TS 36.508 subclause 4.5.2.3 are performed on Cell 2.

    //NOTE: The UE performs registration and the RRC connection is released.

    f_EUTRA_IdleUpdated(eutra_Cell2, PREAMBLE, STATE2_IDLEUPDATE,  NORMAL);

    //@siclog "Step 19" siclog@

    //UE selects PLMN1

    f_UT_PLMN_Check(UT, "00111");

    //@siclog "Step 20" siclog@

    //Check: Does the UE send an RRCConnectionRequest on Cell 4 after 120 seconds, but before 420 seconds from power on?

    alt {

      [] SRB.receive(car_SRB0_RrcPdu_IND(eutra_Cell4, cr_508_RRCConnectionRequest))

        {

          //* @verdict fail RRCConnectionRequest received on PLMNN2 / Cell 4 within t_HpPlmn_lower time

          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 6.1.1.7a Step 20");

        }

      [] t_HpPlmn_lower.timeout

        {

          if (f_EUTRA_RRC_ConnectionRequest_Common(eutra_Cell4, t_HpPlmn_upper)) {

            //* @verdict pass RRC Connection request received on PLMN2 / Cell 4 after t_HpPlmn_lower time but before t_HpPlmn_upper

            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 6.1.1.7a Step 20");

          }

          else {

            //* @verdict fail No RRCConnectionRequest received

            f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 6.1.1.7a Step 20");

          }

        }

    }

    t_HpPlmn_upper.start(f_EUTRA_SetTimerToleranceMin(eutra_Cell4, nasTimer, 420.0));

    v_KsiValue := f_EUTRA_SecurityKSIasme_Get();

    //@siclog "Step 21 - 25" siclog@

    //Steps 2 to 6 of the generic test procedure in TS 36.508 subclause 6.4.2.7 are performed on Cell 4.

    //NOTE: The UE performs a TAU procedure and the RRC connection is released.

    //Step 21: Send RRC Connection Setup on SRB0 (CCCH) and switch on UL grant assignments:

    f_EUTRA_RRC_ConnectionSetup_Def(v_CellId, -, tsc_RRC_TI_Def);

    // UE shall establish SRB1 acc. to 36.331 cl. 5.3.3.4 and 5.3.9

    // UE enters RRC_CONNECTED state

    // SRB1 is established: no ciphering, no integrity protection for NAS/RRC

    //Step 22: Receive RRCConnectionSetupComplete which includes a TRACKING AREA UPDATE REQUEST message

    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(v_CellId,

                                                       cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),

                                                       cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                         cr_508_TAU_Request(v_EpsUpdate_TypeValue,

                                                                                            v_KsiValue,

                                                                                            cr_DRXparameter_Any('5C'O) ifpresent,

                                                                                            v_AdditionalUpdateType_Expected)));

    v_EPS_ContextStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept

    /* Step 23: send TAU ACCEPT and receive TAU COMPLETE; */

    SRB.send(cas_SRB1_NasPdu_REQ(v_CellId,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                cs_508_TAU_Accept(v_EpsUpdate_TypeValue,

                                                                  f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams),

                                                                  v_TaiList,

                                                                  v_EPS_ContextStatus,

                                                                  v_LAI,

                                                                  v_MSId,

                                                                  f_GetAdditionalUpdateResult(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType, v_ForcedAttach)))));

    /* Step24: receive TAU COMPLETE */

    SRB.receive(car_SRB1_NasPdu_IND(v_CellId,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_TAU_Complete)));

    //Step 25: Release RRC connection

    f_EUTRA_RRC_ConnectionRelease(v_CellId, tsc_RRC_TI_Def);

    //@siclog "Step 26" siclog@

    //Check if PLMN2 is indicated by UE

    f_UT_PLMN_Check(UT, "00121");

    //* @verdict pass PLMN2 is indicated by UE

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 6.1.1.7a Step 26");

    //@siclog "Step 27" siclog@

    //Adjust cell power level according to T2
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT2);

    //@siclog "Step 28" siclog@

    //Check for RRCConnectionRequest on Cell1 / PLMN4

    alt {

      [] SRB.receive( car_SRB0_RrcPdu_IND(eutra_Cell1, cr_508_RRCConnectionRequest))

        {

          //* @verdict fail RRCConnectionRequest received on PLMNN4 / Cell 1 within t_HpPlmn_Upper time

          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 6.1.1.7a Step 28");

        }

      [] t_HpPlmn_upper.timeout

        {

          t_IdleMode_GenericTimer.start;

          f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell1);

          t_IdleMode_GenericTimer.stop;

          //* @verdict pass RRC Connection request received on PLMN4 / Cell 1 after t_HpPlmn_Upper time

          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 6.1.1.7a Step 28");

        }

    }

    //@siclog "Step 29 - 33" siclog@

    //Steps 2 to 6 of the generic test procedure in TS 36.508 subclause 6.4.2.7 are performed on Cell 1.

    //NOTE: The UE performs a TAU procedure and the RRC connection is released.

    f_EUTRA_TrackingAreaUpdate_WithoutRrcConnReq(eutra_Cell1);

    //@siclog "Step 34" siclog@

    //Check if PLMN4 is indicated by UE

    f_UT_PLMN_Check(UT, "00101");

    //* @verdict pass PLMN4 is indicated by UE

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 6.1.1.7a Step 34");

    f_EUTRA_TestBody_Set(false);

    //Switch/power off UE

    f_EUTRA_Postamble(eutra_Cell1, E1_IDLE);

  } //end of f_TC_6_1_1_7a_EUTRA
After change

function f_TC_6_1_1_7a_EUTRA() runs on EUTRA_PTC

  { /* PLMN selection / Periodic reselection / MinimumPeriodicSearchTimer / Single Frequency operation */

    var template (value) CellPowerList_Type v_CellPowerList_AtT1;

    var template (value) CellPowerList_Type v_CellPowerList_AtT2;

    var template (value) CellPowerList_Type v_CellPowerList_AtT3;     

    var EUTRA_CellId_Type v_CellId := eutra_Cell4;

    var EUTRA_ATTACH_Type v_ForcedAttach := NORMAL;

    var NAS_MSG_Indication_Type v_NasInd;

    var NAS_KsiValue v_KsiValue;

    var GutiParameters_Type v_GutiParams;

    var B3_Type v_EpsUpdate_TypeValue := f_GetEPSTAUType(v_ForcedAttach);

    var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus;

    var template AdditionalUpdateType v_AdditionalUpdateType_Expected := f_GetAdditionalUpdateType(v_ForcedAttach);

    var NAS_PlmnId v_PLMN;

    var NAS_Lac v_LAC;

    var TrackingAreaCode v_Tac;

    var template (omit)  LocAreaId v_LAI;

    var template (value) TrackingAreaIdList v_TaiList;

    var template (omit)  MobileIdentity v_MSId := f_GetMSId(v_ForcedAttach);

    timer t_HpPlmn_lower;

    timer t_HpPlmn_upper;

    timer t_IdleMode_GenericTimer := tsc_IdleMode_GenericTimer;

    v_CellPowerList_AtT1 := {                                       

      cs_CellPower(eutra_Cell2, -79),

      cs_CellPower(eutra_Cell4, tsc_ServingCellRS_EPRE)

    };

    v_CellPowerList_AtT2 := {                                   

      cs_CellPower(eutra_Cell2, tsc_ServingCellRS_EPRE),

      cs_CellPower(eutra_Cell4, -79)

    };

    v_CellPowerList_AtT3 := {

      cs_CellPower(eutra_Cell1, -79),                                

      cs_CellPower(eutra_Cell2, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower(eutra_Cell4, tsc_ServingCellRS_EPRE)

    };
    //Initialise all cell, security and mobile parameters

    f_EUTRA_Init(c1);

    //Set maximum cell powel level for Cell 1 & 4 to be used in creation

    f_EUTRA_CellInfo_InitMaxReferencePower ( eutra_Cell1, -79 );

    f_EUTRA_CellInfo_InitMaxReferencePower ( eutra_Cell2, -79 );   

    f_EUTRA_CellInfo_InitMaxReferencePower ( eutra_Cell4, -79 );   

    //Set Sys info and cell info for cell 2, &4 to be used @ creation

    f_EUTRA_CellInfo_SetPLMN_1Entry(eutra_Cell2, cs_HPLMN_001_11);  //PLMN1

    f_EUTRA_CellInfo_SetPLMN_1Entry(eutra_Cell4, cs_HPLMN_001_21);  //PLMN2

    //Set TAC for Cell2, &Cell4

    //TAC for Cell1 is set to 1 in f_EUTRA_Init

    f_EUTRA_CellInfo_SetTAC(eutra_Cell2, tsc_IdleMode_TAC_Cell2);

    f_EUTRA_CellInfo_SetTAC(eutra_Cell4, tsc_IdleMode_TAC_Cell4);

    //Configure and activate all cells

    //Cell power level will be -85 dBm/15kHz & attenuation set so that cell is seen as OFF

    f_EUTRA_CellConfig_Def(eutra_Cell2);

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_CellConfig_Def(eutra_Cell4);

    // initialization of variables

    v_GutiParams:= f_EUTRA_CellInfo_GetGuti(v_CellId);

    v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode(v_CellId);

    v_Tac := f_EUTRA_CellInfo_GetTAC(v_CellId);

    v_PLMN := f_Asn2Nas_PlmnId(v_GutiParams.PLMN_Identity);

    v_LAI := f_GetLAI(v_PLMN, v_LAC, v_ForcedAttach);

    v_TaiList := cds_TAIListNonConsecutive_tlv(v_PLMN, { bit2oct(v_Tac) });

    //Equip UE with a USIM containing default values except for those shown in Table 6.1.1.7a.3.1-2

    f_UT_USIM_Insert(UT, "36.523-1 Table 6.1.1.7a.3.1-2");

    //Preamble to ensure UE is in known state

    f_EUTRA_Preamble(eutra_Cell2, STATE2_IDLEUPDATE);

    f_EUTRA_UE_Detach_SwitchOffUe(eutra_Cell2, RRC_IDLE);

    f_EUTRA_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    //Adjust cell configuration according to T1

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 2" siclog@

    //Power on UE

    t_HpPlmn_lower.start(f_EUTRA_SetTimerToleranceMin(eutra_Cell2, nasTimer, 120.0));

    f_EUTRA_SwitchOnUEandStartIP(eutra_Cell2, CNF_REQUIRED);

    t_HpPlmn_upper.start(f_EUTRA_SetTimerToleranceMax(eutra_Cell2, nasTimer, 420.0));

    //@siclog "Step 3-18" siclog@

    //Steps 2 to 17 of the registration procedure described in TS 36.508 subclause 4.5.2.3 are performed on Cell 2.

    //NOTE: The UE performs registration and the RRC connection is released.

    f_EUTRA_IdleUpdated(eutra_Cell2, PREAMBLE, STATE2_IDLEUPDATE,  NORMAL);

    //@siclog "Step 19" siclog@

    //UE selects PLMN1

    f_UT_PLMN_Check(UT, "00111");

    //@siclog "Step 19A" siclog@

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT2);
    //@siclog "Step 20" siclog@

    //Check: Does the UE send an RRCConnectionRequest on Cell 4 after 120 seconds, but before 420 seconds from power on?

    alt {

      [] SRB.receive(car_SRB0_RrcPdu_IND(eutra_Cell4, cr_508_RRCConnectionRequest))

        {

          //* @verdict fail RRCConnectionRequest received on PLMNN2 / Cell 4 within t_HpPlmn_lower time

          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 6.1.1.7a Step 20");

        }

      [] t_HpPlmn_lower.timeout

        {

          if (f_EUTRA_RRC_ConnectionRequest_Common(eutra_Cell4, t_HpPlmn_upper)) {

            //* @verdict pass RRC Connection request received on PLMN2 / Cell 4 after t_HpPlmn_lower time but before t_HpPlmn_upper

            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 6.1.1.7a Step 20");

          }

          else {

            //* @verdict fail No RRCConnectionRequest received

            f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 6.1.1.7a Step 20");

          }

        }

    }

    t_HpPlmn_upper.start(f_EUTRA_SetTimerToleranceMin(eutra_Cell4, nasTimer, 420.0));

    v_KsiValue := f_EUTRA_SecurityKSIasme_Get();

    //@siclog "Step 21 - 25" siclog@

    //Steps 2 to 6 of the generic test procedure in TS 36.508 subclause 6.4.2.7 are performed on Cell 4.

    //NOTE: The UE performs a TAU procedure and the RRC connection is released.

    //Step 21: Send RRC Connection Setup on SRB0 (CCCH) and switch on UL grant assignments:

    f_EUTRA_RRC_ConnectionSetup_Def(v_CellId, -, tsc_RRC_TI_Def);

    // UE shall establish SRB1 acc. to 36.331 cl. 5.3.3.4 and 5.3.9

    // UE enters RRC_CONNECTED state

    // SRB1 is established: no ciphering, no integrity protection for NAS/RRC

    //Step 22: Receive RRCConnectionSetupComplete which includes a TRACKING AREA UPDATE REQUEST message

    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(v_CellId,

                                                       cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),

                                                       cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                         cr_508_TAU_Request(v_EpsUpdate_TypeValue,

                                                                                            v_KsiValue,

                                                                                            cr_DRXparameter_Any('5C'O) ifpresent,

                                                                                            v_AdditionalUpdateType_Expected)));

    v_EPS_ContextStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept

    /* Step 23: send TAU ACCEPT and receive TAU COMPLETE; */

    SRB.send(cas_SRB1_NasPdu_REQ(v_CellId,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                cs_508_TAU_Accept(v_EpsUpdate_TypeValue,

                                                                  f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams),

                                                                  v_TaiList,

                                                                  v_EPS_ContextStatus,

                                                                  v_LAI,

                                                                  v_MSId,

                                                                  f_GetAdditionalUpdateResult(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType, v_ForcedAttach)))));

    /* Step24: receive TAU COMPLETE */

    SRB.receive(car_SRB1_NasPdu_IND(v_CellId,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_TAU_Complete)));

    //Step 25: Release RRC connection

    f_EUTRA_RRC_ConnectionRelease(v_CellId, tsc_RRC_TI_Def);

    //@siclog "Step 26" siclog@

    //Check if PLMN2 is indicated by UE

    f_UT_PLMN_Check(UT, "00121");

    //* @verdict pass PLMN2 is indicated by UE

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 6.1.1.7a Step 26");

    //@siclog "Step 27" siclog@

    //Adjust cell power level according to T3
    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT3);

    //@siclog "Step 28" siclog@

    //Check for RRCConnectionRequest on Cell1 / PLMN4

    alt {

      [] SRB.receive( car_SRB0_RrcPdu_IND(eutra_Cell1, cr_508_RRCConnectionRequest))

        {

          //* @verdict fail RRCConnectionRequest received on PLMNN4 / Cell 1 within t_HpPlmn_Upper time

          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 6.1.1.7a Step 28");

        }

      [] t_HpPlmn_upper.timeout

        {

          t_IdleMode_GenericTimer.start;

          f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell1);

          t_IdleMode_GenericTimer.stop;

          //* @verdict pass RRC Connection request received on PLMN4 / Cell 1 after t_HpPlmn_Upper time

          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 6.1.1.7a Step 28");

        }

    }

    //@siclog "Step 29 - 33" siclog@

    //Steps 2 to 6 of the generic test procedure in TS 36.508 subclause 6.4.2.7 are performed on Cell 1.

    //NOTE: The UE performs a TAU procedure and the RRC connection is released.

    f_EUTRA_TrackingAreaUpdate_WithoutRrcConnReq(eutra_Cell1);

    //@siclog "Step 34" siclog@

    //Check if PLMN4 is indicated by UE

    f_UT_PLMN_Check(UT, "00101");

    //* @verdict pass PLMN4 is indicated by UE

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 6.1.1.7a Step 34");

    f_EUTRA_TestBody_Set(false);

    //Switch/power off UE

    f_EUTRA_Postamble(eutra_Cell1, E1_IDLE);

  } //end of f_TC_6_1_1_7a_EUTRA
5. Execution Log Files
Intel UE
The Intel XMM7260 passed this test case on Anritsu ME7832L LTE System in LTE FDD band 13. The documentation below is enclosed as evidence of the successful test case run [1]:

In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

